

Integer Factorization with User Input for the Casio fx-5800P Calculator

https://github.com/slugrustle/fx-5800P_progs Version 1

Program WFSUB

```

1 E+1→E:
2 B→Z[E]:
3 Do:
4   D→A:
5   Z[E+11]+1→Z[E+11]:
6   A÷B→D:
7 LpWhile Frac(D)=0:
8 Int(√(A))→C:
9 Return

```

Program Outline

Lines 1–2: Increment number of unique factors and store B as a unique factor.

Lines 3–7: Divide out powers of B from A, incrementing the exponent of B in factor storage.

Variable Descriptions

A: Unfactored part of original input number.

B: Known factor of A.

C: Largest remaining factor in A, assuming A is composite.

D: Temporary storage for $A \div B$.

E: Number of stored unique factors (ignoring powers). $E \in [1, 11]$.

Program FACTOR

```

1 0→DimZ:
2 22→DimZ:
3 "NUMBER"?→F:
4 If F<1 Or F≥1×1010:
5   Then "NUMBER MUST BE ≥1 And <1×1010":
6   Stop:
7 IfEnd:
8 If F≠Int(F):
9   Then "NUMBER MUST BE AN INTEGER":
10  Stop:
11 IfEnd:
12 For 1→E To 22:
13   0→Z[E]:
14 Next:
15 1→Z[1]:
16 1→Z[12]:
17 1→E:
18 F→A:
19 Int(√(A))→C:

```

```

20 2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
21 3→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
22 5→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
23 7→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
24 11→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
25 While 1:
26   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
27   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
28   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
29   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
30   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
31   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
32   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
33   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
34   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
35   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
36   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
37   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
38   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:

```

```

39   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
40   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
41   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
42   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
43   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
44   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
45   B+8→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
46   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
47   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
48   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
49   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
50   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
51   B+8→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
52   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
53   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
54   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
55   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
56   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
57   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
58   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
59   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
60   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
61   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
62   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
63   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
64   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
65   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
66   B+6→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
67   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
68   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
69   B+4→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
70   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
71   B+10→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
72   B+2→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
73   B+10→B:A÷B→D:Frac(D)=0⇒Prog "WFSUB":B>C⇒Goto 1:
74   WhileEnd:

```

```

75   Lbl 1:
76   If A>1:
77       Then E+1→E:
78       A→Z[3]:
79       1→A:
80       1→Z[E+11]:
81   IfEnd:
82   Int(E÷3)→D:
83   E-3×D>0⇒D+1→D:
84   1→C:
85   Lbl 2:
86   Cls:
87   Locate 1,1,F:
88   Locate 12,1,C:

```

```

89   Locate 13,1,"":
90   Locate 14,1,D:
91   3×(C-1)+1→B:
92   Locate 1,2,Z[B]:
93   Locate 11,2,"^(":
94   Locate 13,2,Z[B+11]:
95   Locate 16,2,")":
96   If B+1≤E:
97       Then Locate 1,3,Z[B+1]:
98       Locate 11,3,"^(":
99       Locate 13,3,Z[B+12]:
100      Locate 16,3,")":
101   IfEnd:
102   If B+2≤E:

```

```

103 Then Locate 1,4,Z[B+2]:
104 Locate 11,4,"^(":
105 Locate 13,4,Z[B+13]:
106 Locate 16,4,")":
107 IfEnd:
108 While 1:
109   GetKey→A:
110   If A=34 Or A=73:
111     Then Cls:
112     "DONE":
113     Stop:
114   IfEnd:
115   If A=84 Or A=86 Or A=77 Or A=47:
116     Then C+1→C:
117     C>D⇒1→C:
118     Goto 2:
119   IfEnd:
120   If A=83 Or A=85 Or A=67:
121     Then C-1→C:
122     C<1⇒D→C:
123     Goto 2:
124   IfEnd:
125 WhileEnd

```

Program Outline

Lines 1–2: Set up memory for storing 11 factors in the form $Z[\alpha]^Z[\alpha+11]$ where $\alpha \in [1,11]$.

Lines 3–11: User input of number to be factored.

Lines 12–17: Zero out factor storage and store 1^1 as the first factor.

Lines 18–81: Wheel factorization of input with a basis of 2, 3, 5, and 7.

Lines 82–125: Display factors and original input number.

Variable Descriptions

A: Unfactored part of original input number and identifier of the most recently pressed key.

B: Known factor of A and index of displayed factor.

C: Largest remaining factor in A (assuming A is composite) and number of displayed factor page.

D: Temporary storage for $A \div B$ and number of factor display pages.

E: Number of unique factors in factor storage (ignoring powers). $E \in [1,11]$.

F: Number to be factored (input by user).

Notes

There is only storage for 11 unique factors because $6469693230 = 1 \times 2 \times 3 \times 5 \times 7 \times 11 \times 13 \times 17 \times 19 \times 23 \times 29$ has the most unique factors of any number on the range $[1, 1 \times 10^{10}]$.

Line 4: The fx-5800P can only represent numbers on the range $[-1 \times 10^{10}, 1 \times 10^{10}]$ as exact integers.

Lines 5 and 9: The weird spacing prevents text wrapping from occurring in the middle of a word.

Line 110: Pressing DEL (34) or EXIT (73) ends the program.

Line 115: Pressing ▲ (84), ► (86), + (77), or EXE (47) cycles to the next factor display page.

Line 120: Pressing ◀ (83), ▼ (85), or – (67) cycles to the previous factor display page.

Written in 2018 by Ben Tesch. To the extent possible under law, the author has dedicated all copyright and related and neighboring rights to this software to the public domain worldwide. This software is distributed without any warranty. Published under the CC0 1.0 Universal Public Domain Dedication; see <http://creativecommons.org/publicdomain/zero/1.0/>